Maple-assisted proof of formula for A204712

Robert Israel

19 June 2018

The common permanents of the 2×2 blocks could be either 1 or 2. However, the only way the permanents can all be 2 is if all entries are 1. Thus $a(n) = b(n) + 1$, where $b(n)$ counts the cases where the permanents are all 1.
Note that because the sum of coefficients on the right side of the conjectured recurrence a(n) = 8 a(n-1) +50 a(n-2) -389 a(n-3) -790 a(n-4) +6534 a(n-5) +3836 a(n-6) -45232 a(n-7) +9360 a(n-8) +108928 a(n-9) -82304 a(n-10) is 1, $a(n)$ satisfies it if and only if $b(n)$ satisfies the same recurrence.

There are $2^8 = 256$ configurations for a row. Consider the 256×256 transition matrix T with entries $T_{ij} = 1$ if the bottom row of a 2×8 sub-array could be in configuration i while the top row is in configuration j (i.e. all 2×2 blocks have permanent 1), and 0 otherwise. The following Maple code computes it. Configurations are encoded as 8-element lists, corresponding to binary digits.

```
> Configs:=  [seq(convert(2^8+i,base,2)[1..8],i=0..2^8-1)]:
> Compatible:= proc(i,j)
   if andmap(k -> Configs[i][k]*Configs[j][k+1]+Configs[i][k+1]*
  Configs[j][k] = 1, [$1..7]) then 1 else 0 fi
  end proc:
> T:= Matrix(256,256,Compatible):
```

Thus $b(n) = u\, T^n v$ where u and v are row and column vectors respectively of all 1's. The following Maple code produces these vectors.

```
> u:= Vector[row](256, 1):
  v:= Vector(256, 1):
```

To check, here are the first few entries of our sequence.

```
> TV[0]:= v:
  for n from 1 to 24 do TV[n]:= T . TV[n-1] od:
> A:= [seq(1+ u . TV[n],n=1..24)];
```

$$A := [385, 3473, 28161, 239425, 1992321, 16748161, 140090241, 1174759297, 9838208513, 82449830017, 690711971457, 5787565930753, 48489078457729, 406275347589249, 3403932556101121, 28520064001053825, 238954333647573121, 2002083737371343105, 16774435444775234433, 140544717175076647553, 1177553386564242959873, 9866133530675882329729, 82663387744776658539137, 692595249821690009529601] \quad (1)$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \mapsto 2752561807360\, t + t^{34} - 165\, t^{32} - 128\, t^{31} + 11458\, t^{30} + 14080\, t^{29} - 454329\, t^{28} - 692224\, t^{27} + 11631066\, t^{26} + 20110080\, t^{25} - 204859081\, t^{24} - 384864384\, t^{23} + 2574027717\, t^{22} + 5125001216\, t^{21} - 23560968900\, t^{20} - 48936518144\, t^{19} + 158835307400\, t^{18} + 340476741376\, t^{17} - 791128049436\, t^{16} - 1736912215168\, t^{15} + 2899231175616\, t^{14} + 6482979235072\, t^{13} - 7720223264720\, t^{12} - 17508766716928\, t^{11} + 14588363860672\, t^{10} + 33447821788672\, t^{9} - 18774540033536\, t^{8} - 43408761394176\, t^{7} \quad (2)$$

$+ 15302213473280\, t^6 + 35607348998144\, t^5 - 6849302683648\, t^4 - 16012830998528\, t^3$
$+ 1174012559360\, t^2$

This turns out to have degree 34, but with the lowest coefficient 0. Thus we will have

$0 = u\, P(T)\, T^n\, v = \sum_{i=1}^{34} p_i\, b(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 33, which would hold true for any u and v, after a delay of 1. It seems that with our particular u and v we have a recurrence of order only 10, corresponding to $t-1$ times a factor of P (the $t-1$ is needed to get a recurrence with sum of coefficients 0, so that it is satisfied by $a(n)$ as well as $b(n)$; $b(n)$ will satisfy a recurrence of order 9.

```
> Q:= unapply(normal((t^10-8*t^9-50*t^8+389*t^7+790*t^6-6534*t^5
  -3836*t^4+45232*t^3-9360*t^2-108928*t+82304)/(t-1)),t);
```

$$Q := t \mapsto t^9 - 7\, t^8 - 57\, t^7 + 332\, t^6 + 1122\, t^5 - 5412\, t^4 - 9248\, t^3 + 35984\, t^2 + 26624\, t - 82304 \quad \textbf{(3)}$$

The complementary factor $R(t) = \dfrac{P(t)}{Q(t)}$ has degree 25.

```
> R:= unapply(normal(P(t)/Q(t)),t);
```

$$R := t \mapsto \left(-33443840 - 25082880\, t + t^{24} + 7\, t^{23} - 59\, t^{22} - 474\, t^{21} + 1331\, t^{20} + 13525\, t^{19} - 13089\, t^{18} - 213542\, t^{17} + 17085\, t^{16} + 2067091\, t^{15} + 875951\, t^{14} - 12875968\, t^{13} - 9487050\, t^{12} + 52649060\, t^{11} + 49654060\, t^{10} - 141023716\, t^9 - 152973332\, t^8 + 241028608\, t^7 + 285534360\, t^6 - 246758480\, t^5 - 309924800\, t^4 + 131592384\, t^3 + 171821312\, t^2\right) t \quad \textbf{(4)}$$

Now we want to show that $c(n) = u\, Q(T)\, T^n\, v = 0$ for all $n \geq 1$. This will certainly satisfy the recurrence

$$\sum_{i=1}^{25} r_i\, c(i+n) = \sum_{i=1}^{25} r_i u\, Q(T)\, T^{n+i} v = u\, Q(T)\, R(T)\, T^n\, v = u\, P(T)\, T^n\, v = 0$$

where r_i are the coefficients of $R(t)$. To show all $c(n) = 0$ it suffices to show $c(1) = \ldots = c(24) = 0$.

First we compute $w = u\, Q(T)$, then multiply it with the already-computed $T^n v$.

```
> UT[0]:= u:
  for n from 1 to 9 do UT[n]:= UT[n-1].T od:
  w:= add(coeff(Q(t),t,j)*UT[j],j=0..9):
> seq(w . TV[n],n=1..24);
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \quad \textbf{(5)}$$

This completes the proof.